

Tensor Omics: Team Coding Guidelines

May 21, 2026

Contents

1	Compiler: Always Use the Latest gfortran	3
1.1	Installing on Ubuntu	3
1.2	Using the Project Docker Image	3
2	File Extensions: .F90 vs .f90	3
3	Always Use Double Precision	3
4	Variable and Argument Naming Conventions	4
4.1	Prefixes and Suffixes	4
4.2	Mapping Arrays	4
4.3	Boolean / Selection Arrays	4
5	Routine Organisation: Helper / Main / Alloc Pattern	4
5.1	Layer 1: <code>foo_helper</code> — Core Implementation	4
5.2	Layer 2: <code>foo</code> — Validated Entry Point	5
5.3	Layer 3: <code>foo_alloc</code> — Convenience Allocating Wrapper	5
6	Error Handling	6
6.1	Error Code Categories	6
6.2	Always Initialise <code>ierr</code> First	6
6.3	Return Early on Error	6
6.4	<code>set_err</code> vs <code>set_err_once</code>	7
6.5	Never Silent Errors	7
6.6	Avoid GOTO	7
7	Use of <code>intent</code> and <code>pure</code>	7
7.1	Mandatory <code>intent</code>	7
7.2	Mandatory <code>pure</code> for Core Routines	8
7.3	Optional Arguments	8
8	Parallelisation: <code>do concurrent</code>	8
9	Memory Model: No Copies, No Allocations in Cores	9
9.1	No Allocations in Core Routines	9
9.2	No Defensive Copies	9
9.3	Work Arrays (<code>tmp_</code>)	9
9.4	Pointer Views Over Subarrays	9
9.5	Always Use Pointers, Never Value-Copy Scalars in C Wrappers	9
10	Macro Usage (<code>src/macros.h</code>)	9
10.1	Available Macros	10
10.2	Module-Local Constants via <code>#define</code>	10

11 Derived Types (type): Use with Care	10
12 The f42_utils Utility Module	10
13 The safeguard Module	10
14 C Wrapper Conventions	11
14.1 Naming Conventions	11
14.2 Complete Wrapper Template	11
14.3 Key Rules	12
14.4 Character / String Arguments	12
14.5 When Not to Write a Wrapper	13
14.6 Mandatory Development Workflow	13
15 R / Rcpp / Fortran Interface	13
15.1 Rcpp Wrapper Pattern	14
15.2 R Wrapper Pattern	15
15.3 Key Points	15
16 Python / ctypes / Fortran Interface	16
16.1 Loading the Shared Library	16
16.2 Python Wrapper Pattern	16
16.3 Key Points	17
17 FORD Documentation	18
17.1 Comment Syntax	18
17.2 Module and Subroutine Template	18
17.3 Generating Documentation	18
18 Testing	19
18.1 Where to Test What	19
18.2 Test Naming Convention	19
19 Merge Request and Collaboration Conventions	19
20 Getting Help: The ask-for-help Channel	19
21 Code Snippets	20
21.1 How Snippet Generation Works	20
21.2 Rules for Snippet-Compatible Code	20
22 Quick-Reference Checklist	20

1 Compiler: Always Use the Latest gfortran

Always use the latest stable version of `gfortran` (15+). Newer compiler versions bring improved standard compliance, better auto-vectorization, and improved OpenMP/`do concurrent` support.

1.1 Installing on Ubuntu

```
sudo add-apt-repository ppa:ubuntu-toolchain-r/test
sudo apt-get update
sudo apt-get install gfortran-15
```

1.2 Using the Project Docker Image

If `gfortran-15` is not available on your system, use our pre-built Docker image:

```
# Build the image
docker build -t arch-gfortran -f gfortran.docker .

# Run the build inside the container
docker run -it -v 'pwd':/opt arch-gfortran '/opt/build.sh'
```

Using Docker ensures every team member compiles with the same toolchain version, regardless of operating system.

2 File Extensions: .F90 vs .f90

Fortran source files in this project use *two* extensions with a precise meaning:

- `.F90` (uppercase F): The file is passed through the C preprocessor before compilation. Use this extension whenever the file contains `#include`, `#define`, or any other preprocessor directive (e.g., `#include "macros.h"`).
- `.f90` (lowercase f): Plain Fortran, *no* preprocessor. Use this for files that contain no macros or includes.

Rule: If in doubt, use `.F90`. Do not mix files that use macros with the `.f90` extension, as `gfortran` will silently skip the preprocessor pass and produce cryptic compilation errors.

3 Always Use Double Precision

All real variables and constants must use the ISO Fortran double precision kind `real64` from `iso_fortran_env`. Never use the default `real` kind or literal constants without a kind suffix.

```
use, intrinsic :: iso_fortran_env, only: real64, int32

real(real64) :: x, y
integer(int32) :: n

! Compile-time constants MUST carry the _real64 suffix
x = 3.141592653589793_real64
y = 2.718281828459045_real64
```

Omitting the `_real64` suffix causes the literal to be evaluated as single precision before being promoted, silently losing bits of accuracy.

4 Variable and Argument Naming Conventions

Consistent naming makes the code readable without comments. The following conventions are **mandatory**.

4.1 Prefixes and Suffixes

Pattern	Meaning	Example
<code>n_</code>	Size / count of a dimension	<code>n_genes, n_axes, n_families</code>
<code>i_</code>	Loop / iteration variable	<code>i_gene, i_axis, i_vec</code>
<code>_idx</code>	Calculated / derived index	<code>family_idx, out_idx</code>
<code>tmp_</code>	Work array (not input or output)	<code>tmp_perm, tmp_loess_y</code>
<code>_perm</code>	Permutation vector for <name>	<code>sort_perm, gene_perm</code>
<code>_alloc</code>	Routine that allocates work space	<code>normalization_pipeline_alloc</code>
<code>_helper</code>	Core implementation subroutine	<code>k_means_assign_cluster_helper</code>
<code>_c</code>	C-interoperable wrapper	<code>compute_shift_vector_field_c</code>
<code>_expert</code>	Full-control wrapper with all work arrays exposed	<code>compute_family_scaling_expert_c</code>
<code>ierr</code>	Error code argument (always this name)	<code>integer(int32), intent(out) :: ierr</code>

4.2 Mapping Arrays

Use a descriptive `x.to.y` pattern for mapping arrays:

```
integer(int32), intent(in) :: gene_to_fam(n_genes)
! gene_to_fam(i_gene) = family index of gene i_gene

integer(int32) :: family_idx
family_idx = gene_to_fam(i_gene)
```

4.3 Boolean / Selection Arrays

Use the `_mask` or `_selection` suffix for logical index arrays:

```
logical, intent(in) :: vecs_selection_mask(n_vecs)
logical, intent(in) :: axes_selection(n_axes)
```

5 Routine Organisation: Helper / Main / Alloc Pattern

Every non-trivial computation must be split into three layers. This separation keeps core logic pure and testable, while providing convenient entry points for callers at different levels.

5.1 Layer 1: `foo_helper` — Core Implementation

- Contains the actual algorithm. **Must be pure**.
- Accepts *everything* as explicit arguments: input arrays, output arrays, and any pre-allocated work arrays (`tmp_*`).
- Never allocates memory.
- Never validates input; it is the caller's responsibility to pass valid data.

```

pure subroutine quantile_normalization_helper(expr, n_genes, n_replicates, &
    normalized_expr, rank_means, tmp_col, tmp_perm, ierr)
    integer(int32), intent(in) :: n_genes, n_replicates
    real(real64), intent(in)  :: expr(n_replicates, n_genes)
    real(real64), intent(out) :: normalized_expr(n_replicates, n_genes)
    real(real64), intent(in)  :: rank_means(n_replicates)
    real(real64), intent(out) :: tmp_col(n_genes)           ! work array
    integer(int32), intent(out) :: tmp_perm(n_genes)        ! permutation work array
    integer(int32), intent(out) :: ierr
    ! ... pure implementation ...
end subroutine

```

5.2 Layer 2: foo — Validated Entry Point

- Validates all inputs using the `tox_errors` validation routines.
- Returns early on any validation failure (sets `ierr`).
- Calls the `helper` only after successful validation.
- May or may not be `pure`, depending on side effects (e.g. I/O).

```

subroutine quantile_normalization(expr, n_genes, n_replicates, &
    normalized_expr, rank_means, tmp_col, tmp_perm, ierr)
    integer(int32), intent(in) :: n_genes, n_replicates
    real(real64), intent(in)  :: expr(n_replicates, n_genes)
    real(real64), intent(out) :: normalized_expr(n_replicates, n_genes)
    real(real64), intent(in)  :: rank_means(n_replicates)
    real(real64), intent(out) :: tmp_col(n_genes)
    integer(int32), intent(out) :: tmp_perm(n_genes)
    integer(int32), intent(out) :: ierr

    call set_ok(ierr)
    call validate_dimension_size(n_genes, ierr)
    call validate_dimension_size(n_replicates, ierr)
    call validate_all_in_range_real(expr, n_genes * n_replicates, ierr)
    if (is_err(ierr)) return

    call quantile_normalization_helper(expr, n_genes, n_replicates, &
        normalized_expr, rank_means, tmp_col, tmp_perm, ierr)
end subroutine

```

5.3 Layer 3: foo_alloc — Convenience Allocating Wrapper

- Allocates all required work arrays using `M_ALLOCATE`.
- If the algorithm requires sorted inputs, performs sorting here.
- Calls the validated `foo` routine.
- **Never** `pure` (allocates memory).
- This is the routine used from Python/R wrappers and high-level Fortran callers.

```

subroutine quantile_normalization_alloc(expr, n_genes, n_replicates, &
    normalized_expr, rank_means, ierr)
    use tox_errors, only: ERR_ALLOC_FAIL, is_err, set_err, M_USE_ALLOCATION
    integer(int32), intent(in) :: n_genes, n_replicates
    real(real64), intent(in)  :: expr(n_replicates, n_genes)
    real(real64), intent(out) :: normalized_expr(n_replicates, n_genes)
    real(real64), intent(in)  :: rank_means(n_replicates)
    integer(int32), intent(out) :: ierr

    real(real64), allocatable :: tmp_col(:)
    integer(int32), allocatable :: tmp_perm(:)

    call set_ok(ierr)
    M_ALLOCATE(tmp_col(n_genes))
    M_ALLOCATE(tmp_perm(n_genes))

    call quantile_normalization(expr, n_genes, n_replicates, &
        normalized_expr, rank_means, tmp_col, tmp_perm, ierr)
end subroutine

```

6 Error Handling

All errors are communicated through a single integer output argument `ierr`. The `tox_errors` module defines all error codes and the functions to set and check them.

6.1 Error Code Categories

Range	Category	Example constant
0	Success	ERR_OK
1xx	I/O and file errors	ERR_FILE_OPEN, ERR_READ_DATA
2xx	Input validation	ERR_INVALID_INPUT, ERR_DIM_MISMATCH, ERR_NAN_INF
3xx	Memory	ERR_ALLOC_FAIL, ERR_POINTER_NULL
5xxx	Fortran runtime	ERR_UNIT_NOT_CONNECTED
9xxx	Internal / unknown	ERR_INTERNAL, ERR_UNKNOWN

6.2 Always Initialise `ierr` First

Call `set_ok(ierr)` at the start of every subroutine that uses `ierr`:

```

use tox_errors, only: set_ok, set_err, set_err_once, is_err, ERR_INVALID_INPUT

call set_ok(ierr)    ! ierr = ERR_OK = 0

```

6.3 Return Early on Error

Check `ierr` after every operation that can fail, and return immediately:

```

call validate_dimension_size(n_genes, ierr)
call validate_all_in_range_real(data, n_genes, ierr)
if (is_err(ierr)) return

```

6.4 set_err vs set_err_once

- `set_err(ierr, code)`: Always overwrites `ierr` with `code`.
- `set_err_once(ierr, code)`: Only sets `ierr` if it is currently `ERR_OK`. Preserves the first error encountered when chaining multiple fallible calls.

6.5 Never Silent Errors

Do not ignore `ierr`. If a callee sets an error, the caller must either propagate it or handle it explicitly. Every downstream caller that receives an `ierr` must check it.

6.6 Avoid GOTO

Never use GOTO. It is prohibited in all new code for the following reasons:

- It creates non-local control flow that is impossible to follow, making error paths untraceable.
- It defeats the structured error model built around `ierr` and early returns.
- It prevents the compiler from performing flow analysis, blocking optimizations and `pure` qualification.
- It is incompatible with `do concurrent`, which requires structured concurrency regions.

How to replace GOTO correctly:

Pattern to replace	Correct Fortran replacement
Forward jump to error label	<code>call set_err(ierr, ...); return</code>
Loop exit (GOTO after loop)	<code>exit</code> or <code>cycle</code> inside a <code>do</code> loop
Conditional skip block	<code>if (...) then ... end if</code>
Multi-level exit	Refactor into a subroutine with early <code>return</code>

```
! WRONG
if (n <= 0) goto 99
call compute(n)
99 continue

! CORRECT
if (n <= 0) then
  call set_err(ierr, ERR_INVALID_INPUT)
  return
end if
call compute(n)
```

7 Use of intent and pure

7.1 Mandatory intent

Every argument in every subroutine and function must carry an `intent` attribute:

- `intent(in)`: The routine only reads this argument.
- `intent(out)`: The routine completely overwrites this argument.
- `intent(inout)`: The routine both reads and modifies the argument, and the incoming value matters.

Omitting `intent` disables aliasing checks and is forbidden.

7.2 Mandatory pure for Core Routines

All *helper* and core subroutines must be declared **pure**. A **pure** procedure:

- Has no side effects (no I/O, no global state changes).
- Is eligible for auto-vectorization and threading by the compiler.
- Allows the compiler to schedule calls more aggressively.

Remove **pure** only when truly necessary (e.g., allocation, I/O), and only in the `_alloc` or top-level wrapper layer.

7.3 Optional Arguments

Fortran does not support default values in argument lists. Use the `M_DEFAULT_VAL` macro to assign defaults after a `PRESENT` check:

```
#include "macros.h"
subroutine my_routine(data, n, span, degree)
  real(real64), intent(in), optional :: span
  integer(int32), intent(in), optional :: degree
  real(real64) :: actual_span
  integer(int32) :: actual_degree

  M_DEFAULT_VAL(span, actual_span, 0.7_real64)
  M_DEFAULT_VAL(degree, actual_degree, 2_int32)
```

Never use optional arguments in C wrappers; `bind(C)` is incompatible with them.

8 Parallelisation: do concurrent

Prefer `do concurrent` over plain `do` loops whenever the iterations are independent. This is the primary parallelisation mechanism in this project. Do not use OpenMP pragma loops — let the compiler exploit `do concurrent` through its own vectorisation and threading back-ends.

```
! Preferred: explicit concurrency annotation
do concurrent (i_dim = 1:n_dims) shared(vector, vector_norm)
  vector(i_dim) = vector(i_dim) / vector_norm
end do

! With reduction (summing into a scalar is safe)
squares_sum = 0.0_real64
do concurrent (i_element = 1:n_elements) &
  shared(vec, mean_val) reduce(+:squares_sum)
  squares_sum = squares_sum + (vec(i_element) - mean_val) ** 2
end do
```

When to use `do concurrent`:

- Element-wise array operations.
- Independent distance or dot-product computations over gene/axis indices.
- Any loop where iteration order does not affect correctness.

When NOT to use `do concurrent`:

- Loops with data-dependent control flow across iterations (e.g., early-exit on error found inside a loop — in that case, validate before the loop).
- Sequential algorithms like merge steps of merge-sort.

9 Memory Model: No Copies, No Allocations in Cores

9.1 No Allocations in Core Routines

Core (`_helper`) subroutines must not allocate memory. Allocation is performed exclusively in the `_alloc` wrapper layer. This rule:

- Keeps core routines **pure** and deterministic.
- Allows callers to reuse pre-allocated work arrays across multiple calls, avoiding repeated `malloc/free` cycles over large datasets.

9.2 No Defensive Copies

Avoid creating copies of input arrays. For large omics datasets (tens of thousands of genes, hundreds of tissues), an unexpected copy easily exceeds available memory. Design algorithms to work on slices and views of the original data.

9.3 Work Arrays (`tmp_`)

Work arrays must be pre-allocated by the caller and passed as explicit arguments to core routines. They follow the `tmp_` prefix convention.

```
! tmp_perm holds the reusable sort permutation
integer(int32), intent(out) :: tmp_perm(n_genes)
real(real64),   intent(out) :: tmp_loess_y(n_genes)
```

9.4 Pointer Views Over Subarrays

When working with large matrices in the `_alloc` or pipeline layer, use Fortran pointer re-association to create zero-copy views into existing arrays instead of allocating new storage. This is particularly useful for obtaining column or row slices from an already-allocated output matrix:

```
! Create a transposed pointer view into the output array (no copy)
real(real64), dimension(:, :), pointer :: transposed_view
transposed_view(1:n_genes, 1:n_tissues) => log_transformed_expr
! Now use a column slice as a work pointer
tmp_col_ptr => transposed_view(:, 1)
```

This pattern avoids allocating a separate temporary matrix when the required data already exists in the output buffer.

9.5 Always Use Pointers, Never Value-Copy Scalars in C Wrappers

See [Section 14](#).

10 Macro Usage (`src/macros.h`)

All source files that use macros must start with:

```
#include "macros.h"
```

10.1 Available Macros

Macro	Purpose
<code>M_USE_NULL_VALIDATION</code>	Imports modules needed for <code>c_loc</code> /null checks in C wrappers
<code>M_CHECK_IERR_NON_NULL</code>	Silently returns if <code>ierr</code> pointer is null (always first check)
<code>M_CHECK_NON_NULL(ARG)</code>	Sets <code>ERR_POINTER_NULL</code> and returns if <code>ARG</code> pointer is null
<code>M_DEFAULT_VAL(OPT,LOC,DEF)</code>	Assigns <code>OPT</code> to <code>LOC</code> if present, else assigns <code>DEF</code>
<code>M_ALLOCATE(DECL)</code>	Allocates <code>DECL</code> ; sets <code>ERR_ALLOC_FAIL</code> and returns on failure
<code>M_NAN</code>	IEEE quiet NaN (double)
<code>M_NEG_INF</code>	IEEE negative infinity (double)
<code>M_POS_INF</code>	IEEE positive infinity (double)

10.2 Module-Local Constants via `#define`

Define numerical constants that belong to a single module at the top of the file, after the `#include`:

```
#include "macros.h"
#define CM_LOESS_SPAN_DEFAULT 0.7_real64
#define CM_LOESS_DEGREE_DEFAULT 2_int32
```

The `CM_` prefix (“constant macro”) clearly distinguishes them from regular variables.

11 Derived Types (type): Use with Care

Derived types (`type :: ...`) are a powerful Fortran feature but must be used with restraint in this project:

- **Prohibited** in any subroutine or function that has a C wrapper or is otherwise called from Python or R. C, Python `ctypes`, and Rcpp cannot represent Fortran derived types; passing them across the language boundary is undefined behaviour.
- **Permitted** for pure-Fortran data structures where the extra abstraction genuinely improves readability or correctness (e.g., hash maps, JSON serialisation helpers).
- **Avoid by default** in the main TOX scientific routines, which operate exclusively on plain arrays and matrices.

If you need a compound structure that must cross the language boundary, decompose it into flat arrays and pass its components individually.

12 The `f42_utils` Utility Module

The file `src/f42_utils.F90` contains general-purpose helper functions shared across the project: `mean`, `std_dev`, `norm`, `clamp`, `sort_array`, `angle_between`, `is_close`, `logx`, `compute_edf`, etc.

Rule: Before implementing a new helper function anywhere else in the codebase, check whether `f42_utils` already provides it. If you write a function that is genuinely reusable across modules, add it to `f42_utils` rather than creating a module-local copy. This prevents duplication and keeps utility logic in one place.

```
use f42_utils, only: norm, is_close, logx, mean, std_dev, sort_array
```

13 The `safeguard` Module

Every module that defines C wrappers must use `safeguard`. This module contains compile-time assertions that verify the C and Fortran kind constants are equivalent on the target platform:

```
use safeguard ! guarantees c_int == int32 and c_double == real64
```

If these kinds do not match, the module fails to compile with a descriptive error, preventing silent ABI (Application Binary Interface) mismatch bugs. The ABI is the low-level contract between compiled code: it specifies how function arguments are laid out in memory, what sizes the types occupy, and how values are returned. When the Fortran kind for `real64` (8 bytes) does not match the C kind for `c_double` on a given platform, data passed through a C wrapper would be misinterpreted silently, producing incorrect results without any runtime error.

14 C Wrapper Conventions

C wrappers expose Fortran routines to C, Python, and R. They must follow a strict template to ensure ABI stability, null safety, and zero-copy operation.

14.1 Naming Conventions

- A standard wrapper for function `foo` is named `foo_c`.
- A wrapper that exposes all work arrays for fine-grained control is named `foo_expert_c`.
- The `bind(C, name="...")` string must match exactly the function name used in Python and R.

14.2 Complete Wrapper Template

```
#include "macros.h"

pure subroutine compute_tissue_versatility_c( &
    n_axes, n_vectors, expression_vectors, &
    exp_vecs_selection_index, &
    n_selected_vectors, axes_selection, &
    n_selected_axes, &
    tissue_versatilities, tissue_angles_deg, &
    ierr) bind(C, name="compute_tissue_versatility_c")

use, intrinsic :: iso_c_binding, only: c_int, c_double
use tox_tissue_versatility, only: compute_tissue_versatility
M_USE_NULL_VALIDATION
implicit none

! All scalars and arrays must have the target attribute
integer(c_int), intent(in), target :: n_axes
integer(c_int), intent(in), target :: n_vectors
integer(c_int), intent(in), target :: n_selected_axes
integer(c_int), intent(in), target :: n_selected_vectors
real(c_double), intent(in), target :: expression_vectors(n_axes, n_vectors)
integer(c_int), intent(in), target :: exp_vecs_selection_index(n_vectors)
integer(c_int), intent(in), target :: axes_selection(n_axes)
real(c_double), intent(out), target :: tissue_versatilities(n_selected_vectors)
real(c_double), intent(out), target :: tissue_angles_deg(n_selected_vectors)
integer(c_int), intent(out), target :: ierr

! 1. Always check ierr pointer first (silent return if null)
M_CHECK_IERR_NON_NULL
! 2. Check all other pointers
M_CHECK_NON_NULL(n_axes)
M_CHECK_NON_NULL(n_vectors)
M_CHECK_NON_NULL(n_selected_axes)
M_CHECK_NON_NULL(n_selected_vectors)
M_CHECK_NON_NULL(expression_vectors)
```

```

M_CHECK_NON_NULL(exp_vecs_selection_index)
M_CHECK_NON_NULL(axes_selection)
M_CHECK_NON_NULL(tissue_versatilities)
M_CHECK_NON_NULL(tissue_angles_deg)

! 3. Delegate to the core routine
call compute_tissue_versatility(n_axes, n_vectors, expression_vectors, &
    exp_vecs_selection_index, n_selected_vectors, axes_selection,      &
    n_selected_axes, tissue_versatilities, tissue_angles_deg, ierr)

end subroutine compute_tissue_versatility_c

```

14.3 Key Rules

1. **Use iso_c_binding types exclusively:** `real(c_double)`, `integer(c_int)`, `character(kind=c_char)`.
2. **All arguments must have target:** Required for `c_loc()` used by the null-check macros.
3. **Never use value:** All arguments are passed by reference. This is both the Fortran default and a requirement for zero-copy operation across the boundary.
4. **Always explicit array dimensions:** Do not use assumed-shape `(:)` in `bind(C)` routines.
5. **No computation in wrappers:** The wrapper only validates pointers and delegates to the core.
6. **No optional arguments:** `bind(C)` and optional arguments are incompatible. Override via the `_expert_c` variant.
7. **No derived types:** Pass components as flat scalars and arrays.
8. **Stable symbol names:** The `bind(C, name="...")` string is the ABI surface. Never rename it after publication.
9. **No aliasing:** Never pass the same memory region under two different argument names.

14.4 Character / String Arguments

When the Fortran core uses `character(len=*)`, the C wrapper receives a fixed-length character array plus an explicit length integer. Always use **named type parameters** for character declarations to avoid ambiguity between `len` and `kind`:

```

! CORRECT: named parameters, unambiguous
integer(c_int), intent(in), target :: filename_len
character(len=1, kind=c_char), dimension(filename_len), intent(in), target ::
    filename

! CORRECT: multi-dimensional string arrays (len=1 per character element, kind=
    c_char)
integer(c_int), intent(in), target :: max_str_len, d1, d2, d3
character(len=1, kind=c_char), dimension(max_str_len, d1, d2, d3), intent(in),
    target :: strings_nd

! WRONG: positional parameter character(c_char) means len=c_char=1, kind=default
! This happens to work because c_char==1, but is semantically incorrect.
character(c_char), dimension(str_len) :: label ! avoid this

```

The string length must always be passed as a separate integer argument; there is no mechanism to infer it from the C side. After receiving the array, convert it to a Fortran string before calling the core:

```
character(len=:), allocatable :: fortran_str
call c_char_id_as_string(filename, filename_len, fortran_str, ierr)
if (is_err(ierr)) return
```

14.5 When Not to Write a Wrapper

Only create C wrappers for functionality that is genuinely absent in Python and R. Do not wrap:

- Sorting routines (Python `numpy.sort`, R `order()` already exist).
- Generic I/O or string utilities.

14.6 Mandatory Development Workflow

When implementing new scientific functionality, complete these steps *in order* and include all of them in the same Merge Request:

1. **Core Fortran implementation.** Write the `_helper` pure subroutine in the appropriate module. Add the validation layer (`foo`) and the allocating convenience wrapper (`foo_alloc`).
2. **C wrapper.** Create the `foo_c` (and optionally `foo_expert_c`) subroutine with `bind(C)`, null-pointer checks via the `M_CHECK_*` macros, and `use safeguard`. No computation — only delegation.
3. **Python wrapper.** Add the function to `tensoromics_functions.py` with a `#>` snippet header, docstring, correct array layout conversion (`np.asfortranarray` for 2D), `ctypes.byref` for scalars, `check_err_code`, and `_readonly` on all outputs.
4. **R wrapper.** Add the Rcpp forwarder to the `.cpp` file (with Roxygen2 `//'` block and `// [[Rcpp::export]]`) and the R validation wrapper to `tensoromics_functions.R` (`#>` header, `validate_*` calls, `check_err_code`).
5. **FORD documentation.** Ensure all public procedures have `!>` and `!!` comments. Documentation is built by CI, but verify the output locally for non-trivial API changes.
6. **Tests.** Write Fortran unit tests that cover numerical correctness. Add Python/R tests that verify the function is callable and returns the expected type and shape. AI-generated language-wrapper tests are acceptable when they avoid duplicating work at no loss of correctness coverage.

15 R / Rcpp / Fortran Interface

The R interface is a three-layer stack:

1. **Fortran core** — pure computation, no language knowledge.
2. **C wrapper (`bind(C)`)** — type-safe bridge, pointer validation.
3. **Rcpp layer** — declares the Fortran function in `extern "C"`, allocates R vectors, calls the C wrapper, checks `ierr`, returns a named `List`.
4. **R layer** — input validation and user-facing API; calls the `_rcpp` function.

15.1 Rcpp Wrapper Pattern

```
#include <Rcpp.h>
using namespace Rcpp;

// Step 1: Declare the Fortran C wrapper
extern "C" {
    void compute_tissue_versatility_c(
        const int*    n_axes,
        const int*    n_vectors,
        const double* expression_vectors,
        const int*    exp_vecs_selection_index,
        const int*    n_selected_vectors,
        const int*    axes_selection,
        const int*    n_selected_axes,
        double*       tissue_versatilities,
        double*       tissue_angles_deg,
        int*          ierr
    );
}

//' Compute tissue versatility for gene expression vectors
//'
//' @param expression_vectors Numeric matrix (n_axes x n_vectors), column-major
//' @param exp_vecs_selection_mask Logical vector (n_vectors)
//' @param axes_selection Logical vector (n_axes)
//' @return List with tissue_versatilities, tissue_angles_deg, ierr
// [[Rcpp::export]]
List tox_compute_tissue_versatility_rcpp(
    NumericMatrix expression_vectors,
    LogicalVector exp_vecs_selection_mask,
    LogicalVector axes_selection)
{
    int n_axes      = expression_vectors.nrow();
    int n_vectors   = expression_vectors.ncol();
    int n_sel_axes  = sum(axes_selection);
    int n_sel_vecs  = sum(exp_vecs_selection_mask);

    // Convert logical to int (Fortran receives integer 0/1)
    IntegerVector axes_sel_int = as<IntegerVector>(axes_selection);
    IntegerVector vecs_sel_int  = as<IntegerVector>(exp_vecs_selection_mask);

    NumericVector versatilities(n_sel_vecs);
    NumericVector angles_deg(n_sel_vecs);
    int ierr = 0;

    compute_tissue_versatility_c(
        &n_axes, &n_vectors,
        expression_vectors.begin(),
        vecs_sel_int.begin(),
        &n_sel_vecs,
        axes_sel_int.begin(),
        &n_sel_axes,
        versatilities.begin(),
        angles_deg.begin(),
        &ierr);

    return List::create(
        Named("tissue_versatilities") = versatilities,
```

```

    Named("tissue_angles_deg") = angles_deg,
    Named("ierr")              = ierr);
}

```

15.2 R Wrapper Pattern

Input validation belongs exclusively to the R layer (not Rcpp). The project provides a set of reusable atomic validator functions in `rcpp/error_handling.R` (e.g. `validate_numeric_vector`, `validate_numeric_matrix`, `validate_positive_integer_scalar`, `validate_logical_or_index_vector`, `validate_length_equals_n`, etc.). Always use these instead of writing ad-hoc `stopifnot` checks:

```

#> tox_tissue_versatility:compute_tissue_versatility_c: Compute tissue versatility
#' Compute tissue versatility for selected expression vectors
#'
#' @param expression_vectors Numeric matrix (n_axes x n_vectors).
#' @param vector_selection Logical vector, TRUE for vectors to include.
#' @param axis_selection Logical vector, TRUE for axes to include.
#' @return List with tissue_versatilities and tissue_angles_deg.
tox_calculate_tissue_versatility <- function(expression_vectors,
                                             vector_selection,
                                             axis_selection) {
  # Input Validation using shared validators
  validate_numeric_matrix(expression_vectors)
  n_axes      <- nrow(expression_vectors)
  n_vectors   <- ncol(expression_vectors)
  validate_logical_or_index_vector(vector_selection,
                                   expected_length = n_vectors,
                                   name = "vector_selection")
  validate_logical_or_index_vector(axis_selection,
                                   expected_length = n_axes,
                                   name = "axis_selection")

  # Convert logical to integer for Fortran
  vector_selection <- as.integer(vector_selection)
  axis_selection  <- as.integer(axis_selection)

  # Call the Rcpp forwarder
  result <- tox_calculate_tissue_versatility_rcpp(
    expression_vectors, vector_selection, axis_selection)

  # Check error code using shared handler
  check_err_code(result$ierr)

  # Return structured result
  return(list(
    tissue_versatilities = result$tissue_versatilities,
    tissue_angles_deg    = result$tissue_angles_deg
  ))
}

```

15.3 Key Points

- All Rcpp functions must be documented with Roxygen2 comments (`///` blocks) for automatic `man` page generation.
- All data validation lives in the R wrapper, not Rcpp. Use the shared validators from `rcpp/error_handling.R`.
- `ierr` is always returned in the `List` and checked by the R wrapper via `check_err_code(result$ierr)`.

- Column-major memory layout is the natural Fortran storage order; R matrices are column-major by default — no transposition is needed.
- **Memory ownership.** While R generally prevents external mutation of objects, some in-place data manipulation libraries (e.g. `data.table`) can modify underlying buffers. Do not pass aliases of live R objects to Fortran wrappers if those objects may be mutated elsewhere during the same analysis session.
- Never pass R objects by value to the C wrapper; always use `.begin()` or a pointer to the underlying data.
- **Respect the file format.** When adding new functions to `tensoromics_functions.R`, follow the section structure already present in the file (grouped by module, each function preceded by its `#>` snippet header and Roxygen block). Do not add isolated functions between unrelated blocks.

16 Python / ctypes / Fortran Interface

16.1 Loading the Shared Library

```
import ctypes, numpy as np, os
from error_handling import check_err_code

dll_path = os.path.abspath("build/libtensor-omics.so")
ctypes.CDLL("libgomp.so.1", mode=ctypes.RTLD_GLOBAL)
lib = ctypes.CDLL(dll_path)
```

16.2 Python Wrapper Pattern

```
#> tox_tissue_versatility:compute_tissue_versatility_c
def compute_tissue_versatility(expression_vectors, gene_selection, axis_selection):
    """
    Compute tissue versatility for each selected gene expression vector.

    Args:
        expression_vectors: 2D array (n_axes, n_genes), float64, C-contiguous.
        gene_selection:      1D bool array (n_genes).
        axis_selection:      1D bool array (n_axes).

    Returns:
        tissue_versilities: 1D array, length = count(gene_selection).
        tissue_angles_deg:  1D array, length = count(gene_selection).
    """
    expr = np.asfortranarray(expression_vectors, dtype=np.float64)
    n_axes, n_vectors = expr.shape
    vecs_sel = np.ascontiguousarray(gene_selection.astype(np.int32))
    axes_sel = np.ascontiguousarray(axis_selection.astype(np.int32))
    n_sel_vecs = int(np.sum(gene_selection))
    n_sel_axes = int(np.sum(axis_selection))

    versilities = np.zeros(n_sel_vecs, dtype=np.float64)
    angles_deg = np.zeros(n_sel_vecs, dtype=np.float64)
    ierr = ctypes.c_int(0)

    fn = lib.compute_tissue_versatility_c
    fn.restype = None
    fn.argtypes = [
```



```

        ctypes.POINTER(ctypes.c_int),
            n_axes
        ctypes.POINTER(ctypes.c_int),
            n_vectors
        np.ctypeslib.ndpointer(dtype=np.float64, flags="F_CONTIGUOUS"),
            expression_vectors
        np.ctypeslib.ndpointer(dtype=np.int32, flags="C_CONTIGUOUS"),
            vecs_selection
        ctypes.POINTER(ctypes.c_int),
            n_selected_vectors
        np.ctypeslib.ndpointer(dtype=np.int32, flags="C_CONTIGUOUS"),
            axes_selection
        ctypes.POINTER(ctypes.c_int),
            n_selected_axes
        np.ctypeslib.ndpointer(dtype=np.float64, flags="C_CONTIGUOUS"),
            tissue_versatilities
        np.ctypeslib.ndpointer(dtype=np.float64, flags="C_CONTIGUOUS"),
            tissue_angles_deg
        ctypes.POINTER(ctypes.c_int),
            # ierr
    ]
    fn(ctypes.byref(ctypes.c_int(n_axes)),
        ctypes.byref(ctypes.c_int(n_vectors)),
        expr, vecs_sel,
        ctypes.byref(ctypes.c_int(n_sel_vecs)),
        axes_sel,
        ctypes.byref(ctypes.c_int(n_sel_axes)),
        versatilities, angles_deg,
        ctypes.byref(ierr))

    check_err_code(ierr.value)
    _readonly(versatilities, angles_deg)
    return versatilities, angles_deg

```

16.3 Key Points

- 2D arrays passed to Fortran must be **Fortran-contiguous** (`np.asfortranarray`); 1D arrays can be C-contiguous.
- Scalar parameters must be passed **by reference** using `ctypes.byref(c_int(value))`.
- All output arrays must be marked **read-only** after the call with `_readonly(...)`. This prevents accidental mutation of the output buffer by any subsequent Python operation or third-party library (e.g. `matplotlib`, `scipy`) while the memory logically belongs to the Fortran result. Never pass a live, mutable NumPy array as an input to a wrapper if another thread or library could modify it during the same call.
- Check `ierr` immediately after every call via `check_err_code`.
- The `#>` comment above Python wrapper functions follows the project comment convention for documentation generators.
- **Respect the file format.** When adding new functions to `tensoromics.functions.py`, follow the section structure already present (grouped by module, each function preceded by its `#>` snippet header and docstring). Do not add isolated functions between unrelated blocks.

17 FORD Documentation

FORD (Fortran Documentation Generator) produces HTML documentation directly from structured comments. All public procedures and modules must be documented.

17.1 Comment Syntax

- **!>** — Starts the description of a module, subroutine, or function.
- **!!** — Continues a description on the next line (trail comment after a declaration).
- **!|** — Documents an argument; place this comment **on the line immediately after** the argument declaration, not before it.

17.2 Module and Subroutine Template

```
!> Module for computing expression centroids of gene families.
!! Contains the core scientific kernel. C and language wrappers are defined
!! outside the module for compatibility.
module tox_gene_centroids
  use, intrinsic :: iso_fortran_env, only: real64, int32
  implicit none
contains

  !> Computes the element-wise mean for a given set of gene expression vectors.
  !! Returns a single centroid vector of length n_axes.
  pure subroutine mean_vector(expression_vectors, n_axes, n_genes, &
    gene_indices, n_selected_genes, centroid, ierr)
    integer(int32), intent(in) :: n_axes
    !! Number of axes (dimensions)
    integer(int32), intent(in) :: n_genes
    !! Total number of genes in the expression matrix
    real(real64), intent(in) :: expression_vectors(n_axes, n_genes)
    !! Gene expression matrix (n_axes x n_genes)
    integer(int32), intent(in) :: n_selected_genes
    !! Number of genes to include in the centroid
    integer(int32), intent(in) :: gene_indices(n_selected_genes)
    !! Column indices of selected genes in expression_vectors
    real(real64), intent(out) :: centroid(n_axes)
    !! Output centroid vector (n_axes)
    integer(int32), intent(out) :: ierr
    !! Error code: 0 = success
    ! ... implementation ...
  end subroutine mean_vector

end module tox_gene_centroids
```

Note the use of **!!** (two exclamation marks) to place the documentation comment on the line *after* the declaration. This is the FORD convention used throughout the codebase.

17.3 Generating Documentation

Documentation is generated automatically by the CI/CD workflow on every merge to the main branch. **You do not need to run FORD manually.** However, if you want to preview the output locally before submitting an MR:

```
ford ford.yml          # run from project root
open doc/index.html    # review in browser
```

18 Testing

The project provides a structured test suite framework. See the [test directory README](#) for full details on how to add and run tests.

18.1 Where to Test What

- **Algorithm correctness:** Fortran unit tests only. The Fortran test suite is the single source of truth for numerical correctness.
- **Python / R wrappers:** Test only that the function can be called, that the return value is of the correct type and shape, and that reasonable input produces a non-error result.

18.2 Test Naming Convention

Each test file is named `mod.test_<module_name>.{f90|F90|py|R}`. Each test function is named `test_<description>`.

19 Merge Request and Collaboration Conventions

- **Do not merge your own MR.** Always request at least one reviewer and wait for approval before merging.
- **Always assign experienced developers as reviewers.** Add at least Franz or Aaron as a reviewer on every MR. They have deep knowledge of the codebase and can catch issues early.
- **Do not close review threads yourself.** Only the person who opened a thread (the reviewer) should close it after it has been resolved to their satisfaction.
- **Resolve conflicts before requesting review.** Rebasing onto the target branch and fixing conflicts yourself.
- **Keep MRs focused.** One logical change per MR. Large refactors should be discussed in an issue before starting the MR.
- **Link to the relevant issue** in the MR description.
- **Update FORD documentation** and tests as part of the same MR as the implementation.
- **Check the CI/CD workflow.** After pushing or updating an MR, always verify that the automated workflow (tests, build, documentation generation) passes. If it fails, *fix it yourself* — do not wait to be told. A failing pipeline blocks the merge and means the code is not yet ready for review.

20 Getting Help: The ask-for-help Channel

The ask-for-help channel is the right place for **any question** — not just programming. Whether it is about scientific context, workflow setup, naming decisions, tooling, or general project conventions, there will always be someone available to help. Do not hesitate to ask.

For *technical problems* (especially compilation errors), do not spend more than 30 minutes debugging alone. Post in the channel with:

1. The exact compiler command and full error message (copy-paste, do not paraphrase).
2. The minimal code snippet that triggers the error.
3. What you have already tried.

Many gfortran errors for `bind(C)`, `pure`, or kind-mismatch issues are non-obvious; the team has likely encountered the same problem before.

21 Code Snippets

Snippets for Fortran, Python, and R are **auto-generated** by the script `helper/generate_snippets.py` and written to the `snippets/` directory. You should never edit them by hand.

21.1 How Snippet Generation Works

The script parses:

- **Fortran source files** (`src/**/*.F90`) to generate call snippets for each subroutine inside a module.
- **Python files** (`python/*.py`) and **R files** (`rcpp/*.R`) to generate definition and call snippets for each interfacing function.

Snippets are keyed by the special header comment that must appear immediately before each wrapper function:

```
# Python: snippet header format
#> <fortran_module_name>:<fortran_c_wrapper_name>: <description>
def my_function(...):
    ...
```

```
# R: same format
#> tox_tissue_versatility:compute_tissue_versatility_c: Compute tissue versatility
tox_calculate_tissue_versatility <- function(...) {
    ...
}
```

21.2 Rules for Snippet-Compatible Code

- Every Python function in `tensoromics.functions.py` that wraps a Fortran C wrapper must be preceded by a `#>` header.
- Every R function in `tensoromics.functions.R` that wraps a Fortran C wrapper must be preceded by a `#>` header.
- The `<fortran_module_name>` and `<fortran_c_wrapper_name>` in the header must match exactly what is defined in the Fortran source. The script validates this and will raise an error if they do not exist.
- Module names must start with `tox_` or `f42_`. Helper snippets use the `#> (tox|f42)_helper-<name>: <description>` format.
- To run the generator manually: `python helper/generate_snippets.py` from the project root.

22 Quick-Reference Checklist

Use this checklist when reviewing or submitting any Fortran code contribution:

- ☐ File extension is `.F90` (macros used) or `.f90` (no macros).
- ☐ `#include "macros.h"` is present if any `M_*` macro is used.
- ☐ All `real` variables use `real(real64)`; all constants carry `_real64`.
- ☐ Every argument has `intent(in|out|inout)`.
- ☐ Core (helper) routines are declared `pure`.
- ☐ `do concurrent` is used for independent loops.

- ☐ No `GOTO` anywhere.
- ☐ `ierr` is the last (or only output) argument and is initialised with `set_ok(ierr)`.
- ☐ Every `set_err` / allocation failure path returns immediately.
- ☐ Core routines do not allocate; allocation lives in `_alloc` variants.
- ☐ No unexplained copies of large arrays.
- ☐ C wrappers follow the `_c` / `_expert_c` naming convention.
- ☐ All C wrapper arguments carry `target` and are passed by reference.
- ☐ Null checks use `M_CHECK_IERR_NON_NULL` first, then `M_CHECK_NON_NULL` for all others.
- ☐ No derived types in C-interfaced code.
- ☐ `use safeguard` is present in every module with C wrappers.
- ☐ All utility functions are in `f42_utils` (or re-use existing ones).
- ☐ FORD comments (`!>`, `!|`) are present for all public procedures.
- ☐ Fortran tests cover the new logic.
- ☐ Python/R tests verify call-ability and return type.
- ☐ MR does not self-merge and review threads are not self-closed.